
การตรวจสอบ Input แบบ Analog (A/D)

ระบบ Input / Output ของ ไมโครคอนโทรลเลอร P89C51RD2 ของ Philips นัน้ ตามปรกติแลวจะ
เปนแบบ Logic TTL ซึง่สามารถเชื่อมตอโดยตรกกับ Input / Output แบบ TTL เทานั้น แตในการประยุกต
ใชงานบางอยาง อาจมีความจํ าเปนตองติดตั้งอุปกรณตรวจจับ ซึ่งให Output การท ํางานเปนสัญญาณ
แบบ Analog นัน้ ตามปรกติแลวจะไมสามารถเชื่อมตอสัญญาณแบบ Analog เขากับ CPU ไดโดยตรง ดัง
นัน้จ ําเปนตองจัดหาอุปกรณที่จะทํ าหนาที่เปนสื่อกลางในการรับคาสัญญาณ Input แบบ Analog เพื่อสง
ตอให CPU ไดรับรู โดยสํ าหรับบอรดควบคุมของ ET-ROBOT-RD2 นัน้ ไดจัดเตรียมวงจรสํ าหรับทํ าหนาที่
แปลงสัญญาณ Input แบบ Analog ใหเปนสัญญาณขอมูลแบบติจิตอลขนาด ความละเอียด 12 บิต 
จ ํานวน 4 ชอง Input โดยใชชิพ A/D สํ าเร็จรูป เบอร ADS7841 เปนตัวแปลงสัญญาณ

โดยวงจรการเชื่อมตอของ ADS7841 กับไมโครคอนโทรลเลอร P89C51RD2 ในบอรดควบคุมรถ
หุนยนต ET-ROBOT-RD2 นัน้ เปนดังรูป

รปูแสดง วงจร A/D ทีใ่ชกับบอรด ET-ROBOT-RD2

โดยวงจร A/D นีจ้ะสามารถรับสัญญาณ Analog ไดระหวาง 0-5V จ ํานวน 4 ชองสัญญาณ โดยคา
ผลลัพธของแตละชองจะมีคาความละเอียดขนาด 12บิต (0-4095 Step) เชน ถา Analog Input มีขนาด 0V
จะไดผลลัพธเปน 0 (0000H) แตถา Analog Input มขีนาด 2.5V จะไดผลลัพธเปน 2047 (07FFH) หรือถา
สัญญาณ Analog Input มขีนาดเปน 5V กจ็ะไดผลลัพธเปน 4095 (0FFFH) ดังนี้เปนตน

โดยจะเห็นไดวาลักษณะของสัญญาณของ Analog Input นัน้จะรบัเปนแบบแรงดันเทานั้น โดยคา
ผลลัพธที่ไดจากการแปลงคา Analog Input กจ็ะไดเปนคา Step ดังนัน้ถาตองการนํ าคาผลลัพธที่อานได
จาก A/D มาท ํางานแปลงเปนคาแบบแรงดัน ก็จะตองทํ าการนํ าคา Step ทีไ่ดมาท ําการคํ านวณเพื่อเปลี่ยน
จากคา Step เปนคาแรงดันเสียกอนดังนี้

 ขนาดสูงสุดของสัญญาณ Analog Input = 5V
 ขนาดผลลัพธสูงสุดของการแปลงขอมูลจาก A/D = 4095
 คาแรงดันใน 1 Step ของผลลัพธ = 5V / 4095

= 0.001221001 V

ดังนัน้เมือ่ตองการเปลี่ยนคาผลลัพธที่อานไดจาก A/D ใหเปนคาแรงดัน จะตองนํ าคาผลลัพธนั้นไป
คูณดวย 0.001221001V เพือ่ใหไดผลลัพธเปนคาแรงดันที่ตองการ

สํ าหรับการสั่งอานคาขอมูลจาก A/D ของ ADS7841 ดวยโปรแกรม BASCOM-8051 นัน้ ตาม
ปรกตแิลวจะยังไมมีคํ าสั่งสํ าหรับส่ังอานคาผลลัพธจาก ADS7841 ไวใหใชงานโดยตรง ซึ่งจํ าเปนที่เราจะ
ตองสราง Library คํ าสัง่ขึน้มาใชงานเอง ซึ่งถาตองการใหการทํ างานของโปรแกรมมีความเร็วและเกิดประ
สิทธภิาพสูงสุดนั้น จะตองสราง Library คํ าสั่งดวยภาษาแอสแซมบลี้ เพราะถาสรางเปน Library คํ าสั่งโดย
ใชภาษาเบสิกเองจะไดการทํ างานที่ชาลงไปกวาปรกติอยูพอสมควร

สํ าหรบัในที่นี้ผูเขียนจะขอสรางเปน Library คํ าสัง่ไวใหเรียกใชงานกันเลย โดยผูอานไมจํ าเปนตอง
ท ําความเขาใจกับการทํ างานของภาษาแอสแซมบลี้ก็ได เพียงแตทํ าความเขาใจในขอกํ าหนดและวิธีการ
เรียกใชงานคํ าสั่งใน Library กเ็พยีงพอแลว โดย Library ทีจ่ะสรางขึ้นใชงานกับ ADS7841 ของบอรด ET-
ROBOT-RD2 จะมขีอกํ าหนดดังนี้

• ก ําหนดชื่อสัญญาณที่เชื่อมตอกับ DCLK ของ ADS7841 เปน ADS7841_DCLK
• ก ําหนดชื่อสัญญาณที่เชื่อมตอกับ CS ของ ADS7841 เปน ADS7841_CS
• ก ําหนดชื่อสัญญาณที่เชื่อมตอกับ DIN ของ ADS7841 เปน ADS7841_DIN
• ก ําหนดชื่อสัญญาณที่เชื่อมตอกับ DOUT ของ ADS7841 เปน ADS7841_DOUT
• ตองสรางตัวแปรชนิด Byte เพือ่รองรับการทํ างานของคํ าสั่งในการผานคา Channel ของ A/D

ทีต่องการจะอานคา
• ตองสรางตัวแปรชนิด Word เพือ่รองรับการทํ างานของคํ าสั่งในการผานคาผลลัพธของ A/D

การสราง Library คํ าสั่งเพื่ออานคา A/D จาก ADS7841 ของบอรด ET-ROBOT-RD2

[_ADS7841]
;Input  : R0 ^Variable Save Data(Low of Word)
;       : R1 = Channel(0-3)
;Output : Save Data to Variable
_ADS7841:

SETB {ADS7841_CS}       ;Initial CS
     CLR     {ADS7841_DCLK}     ;Initial DCLK

MOV     A,R1 ;Get Channel
     CJNE A,#0,_CH1_7841

MOV     A,#&B10010111      ;CH0,SGL,POWER
SJMP _START_7841
;

_CH1_7841:
     CJNE A,#1,_CH2_7841

MOV     A,#&B11010111      ;CH1,SGL,POWER
SJMP _START_7841

_CH2_7841:
CJNE A,#2,_CH3_7841
MOV     A,#&B10100111      ;CH2,SGL,POWER
SJMP _START_7841

_CH3_7841:
MOV     A,#&B11100111      ;CH3,SGL,POWER

_START_7841:
     MOV     R2,#8 ;LOOP CONTROL BYTE
     CLR     {ADS7841_CS}
_WRITE_ADS7841:
     RLC     A
     MOV     {ADS7841_DIN},C
     SETB    {ADS7841_DCLK}
     CLR     {ADS7841_DCLK}
     DJNZ    R2,_WRITE_ADS7841

     SETB    {ADS7841_DCLK}    ;Skip Busy
     CLR     {ADS7841_DCLK}

     MOV     R2,#8 ;Loop Read High Byte Data
_READ_HIGH:
     SETB    {ADS7841_DCLK}
     MOV     C,{ADS7841_DOUT}
     CLR     {ADS7841_DCLK}
     RLC     A
     DJNZ    R2,_READ_HIGH
     MOV     DPH,A               ;Save Data High Byte

     MOV     R2,#8 ; Loop Read Low Byte Data

_READ_LOW:
     SETB    {ADS7841_DCLK}
     MOV     C,{ADS7841_DOUT}
     CLR     {ADS7841_DCLK}
     RLC     A
     DJNZ    R2,_READ_LOW
     MOV     DPL,A               ;Save Data Low Byte
     SETB    {ADS7841_CS}        ;DISABLE ADC

     MOV     R2,#4 ; Shift Result
_SHIFT_RESULT:
     MOV     A,DPH
     RRC     A
     MOV     DPH,A
     MOV     A,DPL
     RRC     A
     MOV     DPL,A
     DJNZ    R2,_SHIFT_RESULT

;
     MOV     A,DPL ; Get Result Low Byte

MOV     @R0,A ; Save Low Byte Result
     INC R0 ; Next Point to High Byte

MOV A,DPH ; Get Result High Byte
     ANL     A,#&H0F

MOV @R0,A ; Save High Byte Result
     RET
[END]

แสดง Source Code ใน Library คํ าสั่งของ ASD7841 (ET-7841.LIB)

ซึง่ผูเขียนไดสราง Library ไฟลนี้ไวใหเรียบรอยแลว โดยกํ าหนดชื่อไฟลเปน “ET-7841.LIB”  แตถา
ตองการทดลองสราง Library เองกใ็หพิมพคํ าสั่งภาษาแอสแซมบลีดั้งตัวอยางขางตน จากนั้นก็ทํ าการสั่ง
บันทึกเปนไฟลชื่อ ET-7841.LIB แลวท ําการสั่งคัดลอกไฟลดังกลาวไปไวใน Directory ชื่อ LIB ของ
โปรแกรม BASCOM-8051 กพ็รอมใชงานแลว

โดยในการเรียกใชงานคํ าสั่งใน Library จะใชวิธีการสั่ง Include ไฟล Library เขามาใชงานพรอม
กบัประกาศให BASCOM-8051 ทราบวาเราตองการเรียกใชคํ าสั่งใดใน Library บาง ดังตัวอยาง

$lib "et-7841.lib" ‘เรียกใชงาน Library ไฟล ชื่อ “ET-7541.LIB”
$external _ads7841 ‘เรียกใชคํ าสั่งโปรแกรมยอยใน Library

ซึง่หลงัการประกาศคํ าสั่งดังกลาวขางตนไวในสวนหัวโปรแกรมของภาษาเบสิกแลวก็สามารถเรียก
ใชงานคํ าสั่งใน Library ไดแลว โดยวิธีการใชงานใหศึกษาจากตัวอยางโปรแกรมตอไปนี้

$regfile = "89c51rd.dat"     'ก ําหนดใชงานกับ P89C51RD2(Philips)
$ramstart = 0
$ramsize = 256

$crystal = 36864000           'ก ําหนด XTAL = 18.432MHz แบบ X2 Mode

Config Lcdpin = Pin , Db4 = P2.0 , Db5 = P2.1 , Db6 = P2.2 ,

Db7 = P2.3 , E = P2.4 , Rs = P2.5 ‘ก ําหนดการเชื่อมตอกับ LCD
Config Lcd = 16 * 2 ‘ก ําหนดใชงานกับ LCD ขนาด 16 ตัวอักษร 2 บรรทัด
Cursor Off                    'ไมตองแสดง Cursor LCD

$lib "et-7841.lib" ‘เรียกใชงาน Library ไฟล ชื่อ “ET-7541.LIB”
$external _ads7841 ‘เรียกใชคํ าสั่งโปรแกรมยอยใน Library
Ads7841_dclk Alias P3.3     'ก ําหนดการเชื่อมตอกับ Clock Input
Ads7841_cs Alias P3.4         'ก ําหนดการเชื่อมตอกับ Chips Select
Ads7841_din Alias P3.5        'ก ําหนดการเชื่อมตอกับ Data Input
Ads7841_dout Alias P3.7       'ก ําหนดการเชื่อมตอกับ Data Output

Declare Sub Ads7841(ad_channel As Byte,Ad_result As Word)

Dim Ad_channel As Byte ‘ก ําหนดตัวแปรสํ าหรับผานคา Channel
Dim Ad_result As Word ‘ก ําหนดตัวแปรสํ าหรับผานคาผลลัพธ
Dim Adc_calculate As Single
Dim Adc_display As String * 8

Cls                           'ส่ัง Clear หนาจอ Display
Lcd "DEMO ADS7841 A/D" ‘แสดงขอความที่บรรทัดแรก
Locate 2 , 1 ‘ก ําหนด Cursor ไวที่ตัวอักษรแรกของบรรทัดที่2
Lcd "ADC 0 =     Volt" ‘แสดงขอความที่บรรทัดที่2
Do ‘เร่ิมตนวงรอบการอานคา A/D
  Ads7841 , 0 , Ad_result     'ส่ังอานคา  A/D ของ CH0 ไวในตัวแปร AD_RESULT
  Adc_calculate = Ad_result ‘เปลี่ยนคา Word เปนเลขทศนิยม
  Adc_calculate = Adc_calculate * 0.0012210 'คูณผลลัพธกับ Step
  Adc_display = Fusing(adc_calculate , #.#) ‘แสดงผลทศนิยม 1 หลัก
  Locate 2 , 9 ‘ก ําหนด Cursor แสดงผล
  Lcd Adc_display ‘แสดงคาผลลัพธเปน Volt
Loop ‘กลับไปอานคา A/D ใหมไมรูจบ



ตัวอยางการควบคุมรถหุนยนต ET-ROBOT RD2 ดวยภาษาเบสิก BASCOM-8051    

การตรวจสอบ Input แบบ Analog -6- Version 1.00 (เมษายน 2547)

Sub Ads7841(ad_channel As Byte , Ad_result As Word)’โปรแกรมอาน A/D
    $asm

      mov r0,#{ad_result}  ;ให R0 ชีตํ้ าแหนงแอดเดรสไบทต่ํ าของตัวแปรที่เก็บผลลัพธ
      mov r1,{ad_channel}  ;ให R1 = หมายเลขชอง A/D ทีต่องการอานคา (0-3)
      Lcall _ads7841   ;เรียกใชโปรแกรมยอยใน Library เพือ่อานคาผลลัพธ
    $end Asm

End Sub   ‘จบโปรแกรมยอย

แสดง ตัวอยางโปรแกรมอานคา A/D โดยใช Library คํ าสั่งที่สรางขึ้น

สํ าหรบัการทํ างานของโปรแกรมตัวอยางจะเปนการสั่งอานคา A/D ชอง0 แลวนํ าคามาแสดงผลที่
หนาจอ LCD ในรูปแบบของ Volt Meter โดยแสดงคาผลลัพธในหนวยของ Volt โดยการท ํางานสวนที่เกี่ยว
ของกับ Library (ET-7841.LIB) นัน้จะเริ่มจากการสั่ง Include ไฟล “ET-7841.LIB” ข้ึนมาใชงานใน
โปรแกรมซึ่งจะตองนํ าไฟลดังกลาวไปใสไวใน Directory ชื่อ LIB เสยีกอน เนื่องจากเมื่อพบคํ าสั่ง $LIB
โปรแกรม BASCOM-8051 จะไปคนหาไฟลจาก Directory นีเ้สมอ จากนั้นจะเปนการประกาศเรียกใชคํ า
ส่ัง _ADS7841 จาก Library ดังกลาว

โดยจะเหน็ไดวาในสวนของโปรแกรมหลักจะมีการประกาศ Pin Port สํ าหรับใชในการเชื่อมตอของ
CPU กับ ADS7841 ไวในสวนหัวโปรแกรมดวย เนื่องจาก Library จะถกูออกแบบใหมีความออนตัวในการ
น ําไปใชงาน สามารถเปลี่ยนแปลง Pin Port ในการเชื่อมตอไดเอง

จากนัน้กจ็ะมกีารประกาศสรางโปรแกรมยอยสํ าหรับเปนใชทํ าหนาที่เปนตัวกลางในการผานคา
พารามเิตอรระหวางโปรแกรมหลักและคํ าสั่งใน Library โดยจะมกีารสรางตัวแปรขึ้นมารองรับการทํ างาน
ของคํ าสั่งใน Library จ ํานวน 2 ตัวแปร คือ

• AD_CHANNEL เปนตัวแปรแบบ Byte ใชสํ าหรับผานคา Channel ของ A/D ทีต่องการสั่งอาน
คาผลลพัธ ซึ่งสามารถผานคาเปนตัวเลขระหวาง 0-3

• AD_RESULT เปนตัวแปรแบบ Word ใชสํ าหรับผานคาผลลัพธที่ไดจากการอานคา A/D กลับ
มายังโปรแกรม

โดยในสวนของโปรแกรมยอยที่สรางขึ้นมาใชงานนั้น ในการเรียกใชแตละครั้งจะตองมีการผานคา
ตัวแปรจํ านวน 2 คาใหกับโปรแกรมยอยดวย โดยคาแรก (AD_CHANNEL) จะเปนคาตัวเลข 0-3 เพื่อ
ก ําหนดชองของ A/D ทีต่องการอานคา สวนคาที่2 (AD-RESULT) จะเปนตัวแปรสํ าหรับใชเก็บผลลัพธที่ได
จาก A/D ของชองที่กํ าหนด

โดยในการผานคาใหกับคํ าสั่งใน Library นั้นจะใชคํ าสั่งของภาษาแอสแซมบลี้ทั้งหมด โดยจะ
ก ําหนดใหรีจิสเตอร R0 ชีไ้ปที่ตํ าแหนงแอดเดรสของหนวยความจํ าไบทต่ํ าที่ถูกใชเปนตัวแปรสํ าหรับเก็บคา
ผลลัพธ สวนรีจิสเตอร R1 จะใชผานคาที่ผานมาใหกับตัวแปร AD_CHANNEL ไว จากนั้นจึงทํ าการเรียกใช
โปรแกรมยอย ADS7841 ใน Library โดยการทํ างานของโปรแกรมยอยหรือคํ าสั่งใน Library นี้ จะทํ าใหได
ผลลัพธเก็บไวในตัวแปร ADRESULT ทันที

โดยในสวนของโปรแกรมหลักนั้นจะใชคํ าสั่ง DO..LOOP สํ าหรับทํ าหนาที่วนรอบอานคา A/D โดย
จะเรียกใชโปรแกรมยอยที่สรางขึ้น ดังนี้

Ads7841 , 0 , Ad_result     'ส่ังอานคา  A/D ของ CH0 ไวในตัวแปร AD_RESULT

ซึง่บรรทัดคํ าสั่งดังกลาวจะเปนการเรียกใชโปรแกรมยอย ADS7841 โดยมีการผานคาตัวเลข ศูนย
ไปใหกับโปรแกรมยอย ซึ่งหมายถึง A/D ชอง 0 ซึ่งสามารถเปลี่ยนเปนคาตัวเลข 1 หรือ 2 หรือ 3 แทนได
สวนตัวแปรที่2 จะเปนการผานคาตัวแปรชื่อ AD_RESULT สํ าหรับเก็บคาผลลัพธของ A/D ซึ่งเมื่อโปรแกรม
ท ํางานตามบรรทัดคํ าสั่งนี้แลวจะไดคาผลลัพธเก็บไวในตัวแปร AD_RESULT ทันที โดยคาในตัวแปรจะมี
คาระหวาง 0-4095 ซึง่เปนคา Step ของ A/D ทีอ่านได

โดยในโปรแกรมตัวอยางจะทํ าการนํ าคานั้นไปแปลงเปนคาแรงดันเสียกอน โดยนํ าไปคูณกับคาคง
ที่ 0.0012210 เพือ่เปลี่ยนเปนคาแรงดันตามตองการ จากนั้นจะใชคํ าสั่ง Fusing เพื่อจัดรูปแบบการแสดง
ผลคาตวัเลขทศนิยมใหแสดงผลทศนิยมเพียงหลักเดียว

ซึง่ในการนํ าไปประยุกตใชงานนั้น อาจไมจํ าเปนตองทํ าการคํ านวณคาผลลัพธของ A/D ใหเปนคา
แรงดันก็ได แตอาจใชวิธีการเปรียบเทียบคา Step ที่อานไดจาก A/D วาเทากับหรือ มากกวา หรือนอยกวา
คาอางอิงที่มีอยูเพื่อใชประกอบการตัดสินใจแทนก็ได

ตัวอยางโปรแกรมการอานคา A/D โดยไมตองแยกไฟล Library

$regfile = "89c51rd.dat"     'ก ําหนดใชงานกับ P89C51RD2(Philips)
$ramstart = 0
$ramsize = 256

$crystal = 36864000           'ก ําหนด XTAL = 18.432MHz แบบ X2 Mode

Config Lcdpin = Pin , Db4 = P2.0 , Db5 = P2.1 , Db6 = P2.2 ,

Db7 = P2.3 , E = P2.4 , Rs = P2.5 ‘ก ําหนดการเชื่อมตอกับ LCD
Config Lcd = 16 * 2 ‘ก ําหนดใชงานกับ LCD ขนาด 16 ตัวอักษร 2 บรรทัด
Cursor Off                    'ไมตองแสดง Cursor LCD

Ads7841_dclk Alias P3.3     'ก ําหนดการเชื่อมตอกับ Clock Input
Ads7841_cs Alias P3.4         'ก ําหนดการเชื่อมตอกับ Chips Select
Ads7841_din Alias P3.5        'ก ําหนดการเชื่อมตอกับ Data Input
Ads7841_dout Alias P3.7       'ก ําหนดการเชื่อมตอกับ Data Output

Declare Sub Ads7841(ad_channel As Byte,Ad_result As Word)

Dim Ad_channel As Byte ‘ก ําหนดตัวแปรสํ าหรับผานคา Channel
Dim Ad_result As Word ‘ก ําหนดตัวแปรสํ าหรับผานคาผลลัพธ
Dim Adc_calculate As Single
Dim Adc_display As String * 8

Cls                           'ส่ัง Clear หนาจอ Display
Lcd "DEMO ADS7841 A/D" ‘แสดงขอความที่บรรทัดแรก
Locate 2 , 1 ‘ก ําหนด Cursor ไวที่ตัวอักษรแรกของบรรทัดที่2
Lcd "ADC 0 =     Volt" ‘แสดงขอความที่บรรทัดที่2
Do ‘เร่ิมตนวงรอบการอานคา A/D
  Ads7841 , 0 , Ad_result     'ส่ังอานคา  A/D ของ CH0 ไวในตัวแปร AD_RESULT
  Adc_calculate = Ad_result ‘เปลี่ยนคา Word เปนเลขทศนิยม
  Adc_calculate = Adc_calculate * 0.0012210 'คูณผลลัพธกับ Step
  Adc_display = Fusing(adc_calculate , #.#) ‘แสดงผลทศนิยม 1 หลัก
  Locate 2 , 9 ‘ก ําหนด Cursor แสดงผล
  Lcd Adc_display ‘แสดงคาผลลัพธเปน Volt
Loop ‘กลับไปอานคา A/D ใหมไมรูจบ

Sub Ads7841(ad_channel As Byte , Ad_result As Word)’โปรแกรมอาน A/D
  $ASM

     MOV R0,#{AD_RESULT}  ;ให R0 ชีตํ้ าแหนงแอดเดรสไบทต่ํ าของตัวแปรที่เก็บผลลัพธ
     MOV R1,{AD_CHANNEL}  ;ให R1 = หมายเลขชอง A/D ทีต่องการอานคา (0-3)

 SETB {ADS7841_CS}       ;Initial CS
     CLR     {ADS7841_DCLK}     ;Initial DCLK

MOV     A,R1 ;Get Channel
     CJNE A,#0,_CH1_7841

MOV     A,#&B10010111      ;CH0,SGL,POWER
SJMP _START_7841
;

  _CH1_7841:
     CJNE A,#1,_CH2_7841

MOV     A,#&B11010111      ;CH1,SGL,POWER
SJMP _START_7841

  _CH2_7841:
CJNE A,#2,_CH3_7841
MOV     A,#&B10100111      ;CH2,SGL,POWER
SJMP _START_7841

  _CH3_7841:
MOV     A,#&B11100111      ;CH3,SGL,POWER

  _START_7841:
     MOV     R2,#8 ;LOOP CONTROL BYTE
     CLR     {ADS7841_CS}
  _WRITE_ADS7841:
     RLC     A
     MOV     {ADS7841_DIN},C
     SETB    {ADS7841_DCLK}
     CLR     {ADS7841_DCLK}
     DJNZ    R2,_WRITE_ADS7841

     SETB    {ADS7841_DCLK}    ;Skip Busy
     CLR     {ADS7841_DCLK}

     MOV     R2,#8 ;Loop Read High Byte Data
  _READ_HIGH:
     SETB    {ADS7841_DCLK}
     MOV     C,{ADS7841_DOUT}
     CLR     {ADS7841_DCLK}
     RLC     A
     DJNZ    R2,_READ_HIGH
     MOV     DPH,A               ;Save Data High Byte

     MOV     R2,#8 ; Loop Read Low Byte Data
  _READ_LOW:

     SETB    {ADS7841_DCLK}
     MOV     C,{ADS7841_DOUT}
     CLR     {ADS7841_DCLK}
     RLC     A
     DJNZ    R2,_READ_LOW
     MOV     DPL,A               ;Save Data Low Byte
     SETB    {ADS7841_CS}        ;DISABLE ADC

     MOV     R2,#4 ; Shift Result
  _SHIFT_RESULT:
     MOV     A,DPH
     RRC     A
     MOV     DPH,A
     MOV     A,DPL
     RRC     A
     MOV     DPL,A
     DJNZ    R2,_SHIFT_RESULT

;
     MOV     A,DPL ; Get Result Low Byte

MOV     @R0,A ; Save Low Byte Result
     INC R0 ; Next Point to High Byte

MOV A,DPH ; Get Result High Byte
     ANL     A,#&H0F

MOV @R0,A ; Save High Byte Result
     RET

  $END ASM

End Sub   ‘จบโปรแกรมยอย


